
ABSTRACT

Managed Care Insurance has over 60 hospital
and/or physician networks nationwide. As the
number of networks grows, so does the necessity to
develop graphic portrayal of the networks. PC
SAS/GRAPH is used to produce maps of the
networks by county for individual states as well as
by national level. Benefits are realized through cost
savings from the elimination of overlapping
networks, and greater customer satisfaction due to
expanding network services to contingent counties.

The project started with the VP’s request “would
you, could you…”.

INTRODUCTION

Managed Care Insurance spent the past 30 years
developing its reputation for innovative products and
unsurpassed customer service. Their commitment to
managed care dates back to the mid 1980’s when it
contracted with its first network of physicians and
hospitals. By the end of 1987, it had arrangements
with 8 networks, and by 1992, it had  over 60
contracted networks.

As the number of networks increased, manually
tracking the arrangements in the various counties
and states became cumbersome. As additional
networks were added, duplicate services became
more of a possibility.

SAS/GRAPH for the PC, with its state and county
mapping capability was selected for its ease of use.

PROCEDURE

Data files containing managed care networks by
county were created from production batch files
resident on the IBM mainframe using SAS/BASICS.
The data files were downloaded via IRMAWIN to a
PC with SAS/BASICS PC and SAS/GRAPH PC
software.

A generic PC SAS program produces unique yet
standardized maps.

See the APPENDIX for the program.

DISCUSSION

Three major issues are addressed in this paper: (1)
show only the major cities; (2) county names needed
coordinates, and (3) differentiation of counties by
type of managed care service.

1. Show only major cities with clearly labeled
names on the maps.

 Major cities were selected using the variable POP
on the SAS/GRAPH map data set USCITY.  POP
represents the 1990 census population in
thousands. It was used in this application only as a
relative size indicator, not for analysis.. A minimum
population of 100,000 was chosen to define a major
city. However, the minimum population was lowered
to 50,000  for some states with low population
concentrations.

The function SUBSTR was used to eliminate any
cities with names beginning with “WEST”, “EAST”,
“SOUTH” or “NORTH”.

 In order to comply with the clearly labeled
requirement, the variables X and Y, which are used
to plot the city location, were amended. Based on
the final map size (5”x7”) the optimum tolerance for
overlapping cities was determined to be .01.
Therefore, the X,Y coordinates were rounded to the
nearest hundredth resulting in two new variables
named X3, Y3.

The city data set was then sorted by the rounded
coordinates X3, Y3 and POP. LAST.BYVARIABLE
processing was used to select the city with the
highest population within the coordinate “block” of
X,Y rounded to hundredths.

2. Label counties with county names.

SAS/GRAPH 6.04 does not have a map data set
with county center coordinates to be used with the
SAS/GRAPH map data set CNTYNAME. In order to
locate the county center, it was determined that it
would be defined as the center of the polygon
formed by the boundaries of the county. The
coordinates for the counties are contained on the
SAS/GRAPH map data set USCOUNTY and
COUNTY. The SAS/GRAPH map data set
USCOUNTY was selected for this calculation. In
order to determine the center, first the STATE,
COUNTY, X, Y and SEGMENT variables are
selected. Then PROC GREDUCE is applied to the
data, using as ID variables, STATE, COUNTY, and
SEGMENT. The output data set contains all
coordinates required to plot the county polygon with
the variable DENSITY added. DENSITY describes
the importance of the coordinate to the clarity of the
county’s shape. A detailed county outline would
appear, if all points with a DENSITY of 0 through 6

are plotted. However, if only the points with a
densiity of 3 are plotted, then the county’s shape
would be vaguely similar, without detail. Trial and
error found that the best density for calculating the
county center was 4.  PROC SUMMARY is run on
the data set with the subset of all points with a
density less than 4. The resulting single coordinate
X,Y is used as the county center for purposes of
labeling the county’s name.

3. Differentiate counties by multiple managed
care services.

The third issue was to  differentiate between which
counties had hospital agreements only, physician
agreements only, or both service agreements. The
hospital only counties were to be a solid color, while
the physician only counties were to have an open
pattern of lines. If  a county had both services, then
the county would  have an open pattern of lines
“overlaying”  a solid base color .

Again,  consider the county a closed polygon. In
order to create the physician polygon(lines) that
overlay the  hospital polygon(solid), PROC
GREMOVE  removed  the county borders and
combined the  polygon areas.  PROC GREMOVE
was run  against a subset of data created by
matching physician counties with USCOUNTY.  This
created a new map data set of county coordinates
with the same coordinates as the original data set.
However, SAS considers this new data set as
unique, so both polygons can be plotted , one on top
of the other. The result was a map showing the
various services available by county.

CONCLUSION

PC SAS GRAPH was able to satisfy the business
needs of management and  the creative needs of
the programmer.

APPENDIX

*6/25/99=====SESUGMAP.SAS===============;
%LET ST=CT;
%LET STNW=CTHC;
%LET NW=SELECTIVES;
FILENAME GSASFILE "C:\SASWIN\GMS&STNW..CGM";
LIBNAME MAPLIB    "C:\SASWIN\MAPS";
LIBNAME GDEVICE0  "C:\SASWIN\GRAPH\SASHELP";
GOPTIONS GACCESS=GSASFILE GSFLEN=80
DEVICE='CGMFL2C' GSFMODE=REPLACE; RUN;

**** READ IN MARKETING COUNTIES;
OPTIONS PS=58 LS=80 NOCENTER NUMBER;
DATA CLCTPHN(KEEP= STATE COUNTY SEGMENT SYSTEM)
     CLCTPPO(KEEP= STATE COUNTY SEGMENT SYSTEM);
  INFILE CARDS;
    INPUT @1 ST $2. @4 COUNTY 4. @9 PPOPHN $3.
          @13 NETNAME $15.;
      IF INDEX(NETNAME,"&NW");
      IF ST="&ST";
      SEGMENT=1;
      STATE=STFIPS(ST);
      IF PPOPHN='PHN' THEN DO;
        SYSTEM=COUNTY;
        OUTPUT CLCTPHN;END;
      IF PPOPHN='PPO' THEN DO;
        SYSTEM=SUM(COUNTY,(1000));
        OUTPUT CLCTPPO;END;
     CARDS;
CT 1    PPO SELECTIVES
CT 1    PHN SELECTIVES
CT 3    PPO SELECTIVES
CT 3    PHN SELECTIVES
CT 5    PHN SELECTIVES
CT 7    PPO SELECTIVES
CT 9    PPO SELECTIVES
CT 9    PHN SELECTIVES
CT 11   PPO SELECTIVES
CT 11   PHN SELECTIVES
CT 13   PPO SELECTIVES
CT 13   PHN SELECTIVES
CT 15   PPO SELECTIVES
CT 15   PHN SELECTIVES; RUN;
  PROC SORT NODUPKEY DATA=CLCTPHN;
    BY STATE SEGMENT COUNTY; RUN;
  PROC SORT NODUPKEY DATA=CLCTPPO;
    BY STATE SEGMENT COUNTY; RUN;

**** SELECT COUNTY COORDINATES;
DATA CNTYMAP (KEEP= STATE COUNTY X Y SEGMENT);
  SET MAPLIB.USCOUNTY (KEEP= STATE COUNTY X Y
SEGMENT);
    ST=FIPSTATE(STATE);
    IF ST="&ST" AND SEGMENT=1;
    OUTPUT CNTYMAP; RUN;
  PROC SORT DATA=CNTYMAP;
    BY STATE SEGMENT COUNTY;RUN;
  PROC GREDUCE DATA=CNTYMAP OUT=CNTYRED NORETRY;
    ID STATE COUNTY SEGMENT;RUN;



**** CALCULATE COUNTY LABEL COORDINATES;
DATA CNTYSNK;
  SET CNTYRED;
    IF DENSITY < 4; RUN;
  PROC SUMMARY NWAY MISSING DATA=CNTYSNK;
    CLASS STATE COUNTY SEGMENT;
    VAR X Y; OUTPUT OUT=SUMPOINT MEAN=; RUN;
  PROC SORT DATA=SUMPOINT; BY STATE COUNTY; RUN;

**** SELECT CITY LABELS/COORDINATES;
DATA READCITY
    (KEEP= STYLE SIZE Y3 X3 STATE POSITION COLOR
          X Y YSYS XSYS FUNCTION TEXT WHEN POP);
  LENGTH COLOR FUNCTION $8;
  SET MAPLIB.USCITY;
    ST=FIPSTATE(STATE);
    IF ST="&ST";
    IF POP GE 100;
    COLOR='RED';  FUNCTION='LABEL';
    WHEN='A';
    XSYS='2'; YSYS='2';
    X3=ROUND(X,.01); Y3=ROUND(Y,.01);
    TEXT=CITY; POSITION='5';
    STYLE='HWCGM002'; SIZE=.70;
    IF INDEX(CITY,'East ' ) |
       INDEX(CITY,'West ' ) |
       INDEX(CITY,'North ') |
       INDEX(CITY,'South ') THEN DELETE;
  OUTPUT READCITY;
  KEEP STYLE SIZE Y3 X3 STATE POSITION COLOR POP
     X Y YSYS XSYS FUNCTION CITY TEXT WHEN; RUN;
  PROC SORT DATA=READCITY; BY X3 Y3 POP; RUN;

**** PREVENT OVERLAPPING CITY NAMES;
DATA CITYLABL;
  SET READCITY;
    BY X3 Y3 POP;
    KEEP='NO';
    IF LAST.Y3 THEN KEEP='YES';
    IF KEEP='NO' THEN DELETE; RUN;

**** SELECT COUNTY LABELS;
DATA READCNTY;
  LENGTH COLOR FUNCTION $8;
  SET MAPLIB.CNTYNAME;
    ST=FIPSTATE(STATE);
    IF ST="&ST";
    XSYS='2'; YSYS='2'; WHEN='A';
    COLOR='BLACK'; FUNCTION='LABEL';
    TEXT=COUNTYNM; RUN;
  PROC SORT DATA=READCNTY;BY STATE COUNTY;RUN;

**** MERGE COUNTY LABELS/COORDINATES;
DATA CNTYLABL
    (KEEP= POSITION SIZE STYLE STATE COLOR X Y
           YSYS XSYS FUNCTION TEXT WHEN COUNTY);
  MERGE READCNTY (IN=C) SUMPOINT(IN=S);
    BY STATE COUNTY;
    IF C;
    STYLE='HWCGM002'; SIZE=.4; POSITION='5';RUN;

**** CALCULATE NETWORK COORDINATES;
DATA PPOMAP;
  MERGE CLCTPPO
        (IN=L KEEP= STATE SEGMENT COUNTY SYSTEM)
        CNTYMAP (IN=Q);
    BY STATE SEGMENT COUNTY; RUN;
  PROC SORT DATA=PPOMAP;
    BY STATE SEGMENT SYSTEM; RUN;
  PROC GREMOVE DATA=PPOMAP OUT=PHYSICAN;
    BY STATE SEGMENT SYSTEM;ID COUNTY;RUN;

**** SET TOG NETWORK COORDINATES;
DATA MOVEIT;
  SET CNTYMAP(IN=M) PHYSICAN(IN=D);
    IF D THEN SYSTEM=SYSTEM;
    ELSE IF M THEN SYSTEM=COUNTY; RUN;
  PROC SORT; BY STATE SEGMENT SYSTEM; RUN;

**** SET TOG NETWORK AREAS;
DATA TOG (KEEP= STATE COUNTY SEGMENT SYSTEM
SERVICE);
  LENGTH SERVICE $9;
  SET CLCTPHN (IN=H) CLCTPPO (IN=P);
    IF H THEN SERVICE='HOSPITAL';
    ELSE IF P THEN SERVICE='PHYSICIAN'; RUN;
  PROC SORT;
    BY STATE SEGMENT SYSTEM SERVICE; RUN;

**** SET TOG CITY AND COUNTY LABELS/COORDINATES;
DATA CITYCNTY;
  SET CNTYLABL (IN=T) CITYLABL (IN=U);
    SEGMENT=1; RUN;

**** SET LEGEND FOR MAP;
  LEGEND1 ACROSS=2 LABEL=NONE
      VALUE= (C=BLUE F=HWCGM002 H=.60 J=L); RUN;

**** PRODUCE MAPS;
PROC GMAP DATA=TOG  MAP=MOVEIT  ALL ;
  ID STATE SEGMENT SYSTEM;
  CHORO SERVICE/DISCRETE
        CEMPTY=BLUE COUTLINE=BLUE
        ANNOTATE=CITYCNTY LEGEND=LEGEND1;
  PATTERN1  C=YELLOW  V=S;
  PATTERN2  C=RED     V=M3N135;
TITLE1 J=C C=RED F=HWCGM009 H=.6 CM  "HEALTHCARE NETWORK";
TITLE3 J=C C=BLUE F=HWCGM002 H=.7 CM "&ST - &NW";
FOOTNOTE1 J=C C=BLUE F=HWCGM002 H=.3 CM "GMS&STNW &SYSDATE";
RUN; QUIT; FILENAME GSASFILE CLEAR;

HEALTHCARE NETWORK

CT - SELECTIVES

HOSPITAL PHYSICIAN

FAIRFIELD

HARTFORD

LITCHFIELD

MIDDLESEX

NEW HAVEN

NEW LONDON

TOLLAND

WINDHAM

Bridgeport

Waterbury

New Haven

Hartford


